Буни соддалаштирамиз

$$U(A_1A_2A_3)=(A_1\wedge A_2)\vee\left[\left(\bar{A}_2\wedge A_1\right)\vee\left(\bar{A}_2\wedge A_3\right)\right]\sim$$

$$\sim\left[\underbrace{\left(A_1\wedge\left(A_2\vee\bar{A}_2\right)\right)}_{A_1}\right]\vee\left(\bar{A}_2\wedge A_3\right)\sim\left(A_1\vee\left(\bar{A}_2\wedge A_3\right)\right)=U(A_1,\bar{A}_2,A_3)$$

Унга мос схема:




Демак, (1) схемани ишлаш принципини аниқлаш учун (2) схемани ишлаш принципидан фойдаланиш кулайроқдир, бунинг учун (2) га мос ростлик жадвали тузамиз, 1-тоқ ўтишини, 0 ток ўтмаслигини билдиради.

| *A1* | *A2* | *A3* | $\bar{A}_2$ | $\bar{A}_2\wedge A_3$ | $U(A_1,\bar{A}_2,A_3)$ |
|---|---|---|---|---|---|
| 0 | 0 | 0 | 1 | 0 | 0 |
| 0 | 0 | 1 | 1 | 1 | 1 |
| 0 | 1 | 0 | 0 | 0 | 0 |
| 1 | 0 | 0 | 1 | 0 | 1 |
| 1 | 1 | 0 | 0 | 0 | 1 |
| 0 | 1 | 1 | 0 | 0 | 0 |
| 1 | 0 | 1 | 1 | 1 | 1 |
| 1 | 1 | 1 | 0 | 0 | 1 |

Умумий имкониятлар сони:

$$C_3^0+C_3^1+C_3^2+C_3^3=1+3+3+1=8 \text{ та}$$

(0,0,1), (0,0,0), (0,1,0) да ток ўтмайди, (0,1,1), (1,0,1),(1,0,0), (1,1,1), (1,1,0) да ток ўтади.

**Адабиётлар:**

1. В.И. Игошин. Математическая логика и теория алгоритмов. Саратов, 1991.
2. Ф.А. Бауэр, Г. Гооз. Информатика. Вводный курс. В 2-х ч. Перев. с нем. М.Мир. 1990
3. М.Г. Коляда. Окно в удивительный мир информатики. Д. Сталкер. 1999.
4. И. Неъматов. Дискрет математика ва математик мантик элементлари. Фарғона, 2014 й.

**Дилобар Султонова**
**(Фергана, Узбекистан)**

## РЕШЕНИЕ ЗАДАЧИ КОШИ ДЛЯ УРАВНЕНИЯ БЕСКОНЕЧНОЙ СТРУНЫ В ПАКЕТЕ MAPLE

Рассмотрим уравнения колебаний бесконечной струны

$$\frac{\partial^2 u(x,t)}{\partial t^2}=a^2\frac{\partial^2 u(x,t)}{\partial x^2} \qquad (1)$$

в плоскости $xOt$. Поставим и исследуем следующую задачу в пакете Maple.

**Задача Коши.** Найти решение уравнения (1), удовлетворяющее начальным условиям

$$u(x,0)=f(x), \qquad (2)$$

$$\left.\frac{\partial u(x,t)}{\partial t}\right|_{t=0}=F(x). \qquad (3)$$

Известно, что данная задача аналитическом методе решено в многих классических литератур, например в [1].

Мы решим эту задачу в пакете Maple [2], [3].

> **restart;**

Данная команда очищает память и необходима, если перед этим производились какие-то вычисления.

Записываем уравнение колебаний, присваивая его «значение» переменной **edu**:

> **edu:=a*a*diff(u(x,t),x$2)=diff(u(x,t),t$2);**

После нажатия клавиши ввода на экране появляется уравнение, записанное в обычной математической форме:

$$edu := a^2\left(\frac{\partial^2}{\partial x^2}\mathrm{u}(x,t)\right)=\frac{\partial^2}{\partial t^2}\mathrm{u}(x,t)$$

Решить это уравнение можно с помощью функции pdsolve:

> **h:=pdsolve(edu,u);**

$$h := \mathrm{u}(x,t)=\_\mathrm{F1}(a\,t+x)+\_\mathrm{F2}(a\,t-x)$$

Запись _F1, _F2 означает, что это произвольные функции. Подставляем в h значение t=0,

> **h1:=subs(t=0,h);**

$$h1 := \mathrm{u}(x,0)=\_\mathrm{F1}(x)+\_\mathrm{F2}(-x)$$

и заменяем переменную $x$ на $\xi$ (это сделано для удобства записи последующих выражений):

> **h2:=subs(x=xi,h1);**

$$h2 := \mathrm{u}(\xi,0)=\_\mathrm{F1}(\xi)+\_\mathrm{F2}(-\xi)$$

Приравниваем правую часть полученного выражения к функции $f(\xi)$, описывающей начальные отклонения точек струны от положения равновесия:

> **eqh1:=rhs(h2)=f(xi);**

$$eqh1 := \_\mathrm{F1}(\xi)+\_\mathrm{F2}(-\xi)=\mathrm{f}(\xi)$$

Выделяем правую часть выражения h

> **h3:=rhs(h);**

$$h3 := \_\mathrm{F1}(a\,t+x)+\_\mathrm{F2}(a\,t-x)$$

и находим производную по *t*:

> **h4:=diff(h3,t);**

$$h4 := \mathrm{D}(\_F1)(a\,t+x)\,a+\mathrm{D}(\_F2)(a\,t-x)\,a$$

Подставляем в h4 значение t=0 и снова заменяем переменную $x$ на $\xi$

> **h2:=subs(t=0,x=xi,h4);**

$$h2 := \mathrm{D}(\_F1)(\xi)\,a+\mathrm{D}(\_F2)(-\xi)\,a$$

Приравниваем найденную производную $\partial u(\xi,t)/\partial t\big|_{t=0}$ к функции $F(\xi)$, определяющей начальные скорости точек струны:

> **eqh2:=h2=F(xi);**

$$eqh2 := \mathrm{D}(\_F1)(\xi)\,a+\mathrm{D}(\_F2)(-\xi)\,a=\mathrm{F}(\xi)$$

Выделяем левую часть полученного равенства

> **eqh3:=lhs(eqh2);**

$$eqh3 := \mathrm{D}(\_F1)(\xi)\,a+\mathrm{D}(\_F2)(-\xi)\,a$$

и интегрируем по $\xi$ в переделах от 0 до $x$:

> **eqh3:=int(eqh3,xi=0..x);**

$$eqh3 := a\,\_\mathrm{F1}(x)-a\,\_\mathrm{F2}(-x)-a\,\_\mathrm{F1}(0)+a\,\_\mathrm{F2}(0)$$

Точно так же поступаем с правой частью:

> **eqh4:=rhs(eqh2);**

$$eqh4 := \mathrm{F}(\xi)$$

> **eqh4:=int(eqh4,xi=0..x);**

$$eqh4 := \int_0^x \mathrm{F}(\xi)\,d\xi$$

Теперь вновь объединяем проинтегрированные части в одно уравнение:

> **eqh2:=eqh3=eqh4;**

$$eqh2 := a\,\_\mathrm{F1}(x)-a\,\_\mathrm{F2}(-x)-a\,\_\mathrm{F1}(0)+a\,\_\mathrm{F2}(0)=\int_0^x \mathrm{F}(\xi)\,d\xi$$

В выражении для **eqh1** возвращаемся к старой переменной:

> **eqh1:=subs(xi=x,eqh1);**

$$eqh1 := \_\mathrm{F1}(x)+\_\mathrm{F2}(-x)=\mathrm{f}(x)$$

Получили два уравнения для нахождения неизвестных функций _F1 и _F2. Объединяем эти уравнения в систему и решаем её:

> **r1:=solve({eqh1,eqh2}, {_F1(x), _F2(-x)});**

$$r1 := \left\{ \_F1(x) = \frac{1}{2}\,\frac{a\,f(x) + a\,\_F1(0) - a\,\_F2(0) + \int_0^x F(\xi)\,d\xi}{a},\ \_F2(-x) = \frac{1}{2}\,\frac{a\,f(x) - a\,\_F1(0) + a\,\_F2(0) - \int_0^x F(\xi)\,d\xi}{a} \right\}.$$

Поскольку в решение уравнения колебаний $h$ входит функция $\_F1(at+x)$, то делаем соответствующую замену переменных

> **r2:=subs(x=a*t+x,r1[1]);**

$$r2 := \_F1(a\,t+x) = \frac{1}{2}\,\frac{a\,f(a\,t+x) + a\,\_F1(0) - a\,\_F2(0) + \int_0^{a\,t+x} F(\xi)\,d\xi}{a}$$

и упрощаем полученное выражение

> **r2:=expand(r2);**

$$r2 := \_F1(a\,t+x) = \frac{1}{2}\,f(a\,t+x) + \frac{1}{2}\,\_F1(0) - \frac{1}{2}\,\_F2(0) + \frac{\frac{1}{2}\int_0^{a\,t+x} F(\xi)\,d\xi}{a}.$$

То же самое делаем с функцией $\_F2$:

> **r3:=subs(x=x-a*t,r1[2]);**

$$r3 := \_F2(a\,t-x) = \frac{1}{2}\,\frac{a\,f(x-a\,t) - a\,\_F1(0) + a\,\_F2(0) - \int_0^{x-a\,t} F(\xi)\,d\xi}{a},$$

> **r3:=expand(r3);**

$$r3 := \_F2(a\,t-x) = \frac{1}{2}\,f(x-a\,t) - \frac{1}{2}\,\_F1(0) + \frac{1}{2}\,\_F2(0) - \frac{1}{2}\,\frac{\int_0^{x-a\,t} F(\xi)\,d\xi}{a}.$$

Записываем решение задачи Коши, подставляя найденные функции _F1 и _F2 в общее решение h:

> **h1:=subs(r3,r2,h);**

$$h1 := u(x,t) = \frac{1}{2}\,f(a\,t+x) + \frac{\frac{1}{2}\int_0^{a\,t+x} F(\xi)\,d\xi}{a} + \frac{1}{2}\,f(x-a\,t) - \frac{1}{2}\,\frac{\int_0^{x-a\,t} F(\xi)\,d\xi}{a}.$$

С помощью функции combine объединяем два интеграла в этом выражении в один:

> **h1:=combine(h1);**

$$h1 := u(x,t) = \int_{x-a\,t}^{a\,t+x} \frac{1}{2}\,\frac{F(\xi)}{a}\,d\xi + \frac{1}{2}\,f(a\,t+x) + \frac{1}{2}\,f(x-a\,t)$$

замечаем, что найденное в пакете Maple решение задачи Коши полностью эквивалентно тому, которое приведено в [1].

**Литература:**


1. Тихонов А.Н., Самарский А.А. Уравнения математической физики. – М.: Наука, 1977. – 736 с.
2. Дьяконов В.П. Maple 6: учебний курс. – СПб.: Питер, 2001.
3. Говорухин В.Н., Цибулин В.Г. Введение в Maple V. Математический пакет для всех. М.: Мир, 1997.